Κεφάλαιο 5ο: Επίλυση εξισώσεων και συστημάτων

5.1 Επίλυση εξισώσεων

Το *Mathematica* διαθέτει αρκετές συναρτήσεις για την επίλυση εξισώσεων. Αυτές είναι:

Solve[eqn, x]	επιλύνει την εξίσωση (ή το σύστημα) eqn με άγνωστο το x.
Reduce[eqn, x]	επιλύνει την εξίσωση (ή το σύστημα) eqn με άγνωστο το x και κάνει διερεύνηση.
NSolve[eqn, x]	επιστρέφει αριθμητικές λύσεις της εξίσωσης eqn.
FindRoot[eqn, {x, x_0}]	επιστρέφει μια προσεγγιστική λύση της εξίσωσης eqn στην περιοχή του σημείου x_0.
Roots[eqn, x]	επιλύνει την πολυωνιμική εξίσωση eqn με άγνωστο το x.

Με τις συναρτήσεις **Solve**, **Reduce** και **Roots** βρίσκουμε ακριβείς λύσεις των εξισώσεων ενώ με τις συναρτήσεις **NSolve** και **FindRoot** βρίσκουμε μόνο αριθμητικές (προσεγγιστικές) λυσεις.

Με ακρίβεια μπορούμε να επιλύσουμε ένα μεγάλο πλήθος διαφορετικών εξισώσεων, αλλά όχι όλα τα είδη. Π.χ. μια πολυωνυμική εξίσωση μέχρι τετάρτου βαθμού επιλύνεται πάντοτε με ακρίβεια, αλλά μια πολυωνυμική εξίσωση βαθμού ανωτέρου του τετάρτου μπορεί να λυθεί με ακρίβεια αλλά μπορεί και όχι.

Προσεγγιστικά μπορούμε να λύσουμε όλα τα είδη των εξισώσεων και συστημάτων.

5.1.1 Ακριβής επίλυση εξισώσεων

Η βασική συνάρτηση επίλυσης εξισώσεων με ακρίβεια είναι η **Solve**. Η εξίσωση στο *Mathematica* εισάγεται με διπλό σύμβολο ισότητας ==, αφού το απλό σύμβολο της ισότητας χρησιμοποιείται από πρόγραμμα για ορισμούς. Οι λύσεις x_0, x_1, ... εμφανίζονται σε λίστα υπό τη μορφή κανόνων αντικατάσταση "→", ως εξής:

$$\{\{x \to x_0\}, \{x \to x_1,\}, \ldots\}.$$

Έστω η εξίσωση τρίτου βαθμού $x^3 + 4x^2 - 11x - 30 = 0$, την οποία αποθηκεύουμε στην μεταβλητή eqn.

```
eqn = x^3 + 4 x^2 - 11 x - 30 == 0;
```

Επειδή στο τέλος της προηγούμενης εντολής υπάρχει το ";", δεν εμφανίζεται το output.

Στη συνέχεια επιλύουμε την εξίσωση eqn1 ως προς x (με τη χρήση της συνάρτησης **Solve**) και τη λύση την αποθηκεύουμε στην μεταβλητή sol.

```
sol = Solve[eqn, x]

{{x → -5}, {x → -2}, {x → 3}}
```

Επαλήθευση μπορούμε να κάνουμε είτε με την εντολή

```
x^3 + 4 x^2 - 11 x - 30 /. sol

{0, 0, 0}
```

είτε με την εντολή.

```
eqn /. sol

{True, True, True}
```

Επειδή συνήθως είναι πιο βολικό να έχουμε τις λύσεις x_0, x_1, ... σε μορφή λίστα {x_0, x_1, ... }, χωρίς κανόνες αντικατάστασης, μπορούμε να χρησιμοποιήσουμε τον τελεστή αντικατάστασης "/." για να το επιτύχουμε. Πολλές φορές, μάλιστα, είναι σκόπιμο να δώσουμε στη λίστα που θα προκύψει ένα όνομα.

```
X = x /. sol

{-5, -2, 3}
```

Με τον τρόπο αυτό, δηλαδή έχοντας τις λύσεις στη διάθεση μας ως λίστα, μπορούμε να κάνουμε πράξεις:

```
X + 5

{0, 3, 8}
```

```
X^2

{25, 4, 9}
```

η να αναφερθούμε σε κάποια συγκεκριμένη λύση, γράφοντας κατά τα γνωστά X[[i]], i = 1, 2, 3 ... Π.χ. για να πάρουμε τη λύση x = -5 γράφουμε

```
X[[1]]

-5
```

Επίσης, με τη βοήθεια της λίστας, η επαλήθευση για κάποια συγκεκριμένη λύση (π.χ. για x = -2) γίνεται είτε με την εντολή:

```
x^3 + 4 x^2 - 11 x - 30 /. x -> X[[2]]

0
```

είτε με την εντολή:

```
eqn /. x → X[[2]]

True
```

Αν μια λύση είναι διπλή (ή τριπλή κ.λ.π.) η Solve την επιστρέφει δύο (ή τρεις κ.λ.π.) φορές στη λίστα των λύσεων, ενώ όταν δεν υπάρχουν λύσεις, επιστρέφει { }.

Τις ίδιες εξισώσεις,που επιλύνει η συνάρτηση **Solve**,επιλύνει και η συνάρτηση **Reduce**. Οι διαφορές ανάμεσα στην συνάρτηση **Reduce** και στην συνάρτηση **Solve** είναι οι εξής:

1. Η συνάρτηση **Reduce** παρουσιάζει τις λύσεις υπό τη μορφή:

$$x == x_0 \,||\, x == x_1 \ldots.$$

(θυμίζουμε ότι || είναι ο λογικός τελεστής, που παριστάνει το διαζευτικό ή).

2. Όταν η εξίσωση περιλαμβάνει μια ή περισσότερες παραμέτρους, η **Reduce** την επιλύνει παρουσιάζοντας όλες τις δυνατές περιπτώσεις, κάνοντας δηλαδή ουσιαστικά διερεύνηση, κάτι που δεν κάνει η **Solve**.

Παράδειγμα 1: Να λυθεί η εξίσωση $x^3 - x^2 - 4 = 0$.

Επίλυση της εξίσωσης με τη συνάρτηση **Solve**:

```
Solve[x^3 - x^2 - 4 == 0, x]
```

$$\left\{\{x \to 2\}, \left\{x \to \frac{1}{2}\left(-1 - i\sqrt{7}\right)\right\}, \left\{x \to \frac{1}{2}\left(-1 + i\sqrt{7}\right)\right\}\right\}$$

Μετατροπή των λύσεων από κανόνες αντικταστάσης σε λίστα:

```
x /. %
```

$\left\{2, \frac{1}{2}\left(-1-i\sqrt{7}\right), \frac{1}{2}\left(-1+i\sqrt{7}\right)\right\}$

Οι λύσεις που παίρνουμε με τη συνάρτηση Solve είναι ακριβείς. Φυσικά μπορούμε να πάρουμε αμέσως αριθμητικές (προσεγγιστικές) λύσεις με τη συνάρτηση **N**, και μάλιστα με όση ακρίβεια δεκαδικών επιθυμούμε:

```
N[%%]
```

```
{{x → 2.}, {x → -0.5 - 1.32288 i}, {x → -0.5 + 1.32288 i}}
```

Επίλυση της εξίσωσης με τη συνάρτηση **Reduce**:

```
lyseis = Reduce[x^3 - x^2 - 4 == 0, x]
```

$x == 2 \,||\, x == \frac{1}{2}\left(-1-i\sqrt{7}\right) \,||\, x == \frac{1}{2}\left(-1+i\sqrt{7}\right)$

Για να πάρουμε τις λύσεις σε μορφή λίστας, εφαρμόζουμε πρώτα την εντολή {**ToRules**[%]}, για να τις εμφανίσουμε ως κανόνες αντικατάστασης, και στη συνέχεια εφαρμόζουμε τον τελεστή αντικταστάσης "/.":

```
kanones = {ToRules[lyseis]}
```

$\left\{\{x \to 2\}, \left\{x \to \frac{1}{2}\left(-1-i\sqrt{7}\right)\right\}, \left\{x \to \frac{1}{2}\left(-1+i\sqrt{7}\right)\right\}\right\}$

```
x /. kanones
```

$\left\{2, \frac{1}{2}\left(-1-i\sqrt{7}\right), \frac{1}{2}\left(-1+i\sqrt{7}\right)\right\}$

Ισοδύναμα μπορούμε να εφαρμόσουμε και τον σύνθετο τελεστή x /. {**ToRules[lyseis]**}:

```
x /. {ToRules[lyseis]}
```

$\left\{2, \frac{1}{2}\left(-1-i\sqrt{7}\right), \frac{1}{2}\left(-1+i\sqrt{7}\right)\right\}$

Εύρεση αριθμητικών (προσεγγιστικών) λύσεων:

```
N[kanones]
```

$\{\{x\to 2.\},\ \{x\to -0.5-1.32288\,i\},\ \{x\to -0.5+1.32288\,i\}\}$

Παράδειγμα 2: Να λυθεί η εξίσωση $a\,x^2 - 2\,x + 4 = 0$.

Επίλυση της εξίσωσης με τη συνάρτηση **Solve**:

```
Solve[a x^2 - 2 x + 4 == 0, x]
```

$$\left\{\left\{x\to \frac{2-\sqrt{4-16\,a}}{2\,a}\right\},\ \left\{x\to \frac{2+\sqrt{4-16\,a}}{2\,a}\right\}\right\}$$

Επίλυση της εξίσωσης με τη συνάρτηση **Reduce**:

```
Reduce[a x^2 - 2 x + 4 == 0, x]
```

$$a == 0\ \&\&\ x == 2\ ||\ x == \frac{2-\sqrt{4-16\,a}}{2\,a}\ \&\&\ a \neq 0\ ||\ x == \frac{2+\sqrt{4-16\,a}}{2\,a}\ \&\&\ a \neq 0$$

Μετατροπή των λύσεων σε κανόνες αντικατάστασης:

```
{ToRules[%]}
```

$$\left\{\{a\to 0,\ x\to 2\},\ \left\{x\to \frac{2-\sqrt{4-16\,a}}{2\,a}\right\},\ \left\{x\to \frac{2+\sqrt{4-16\,a}}{2\,a}\right\}\right\}$$

Βλέπουμε ότι επιλύοντας την εξίσωση με τη συνάρτηση **Solve**, το *Mathematica* τη θεωρεί δευτεροβάθμια, δηλαδή υποθέτει ότι $a \neq 0$ ενώ επιλύοντας την εξίσωση με τη συνάρτηση **Reduce** γίνεται πλήρης διερεύνηση. Συγκεκριμένα, βρέθηκε επιπλέον η λύση $x = 2$, η οποία προκύπτει όταν $a = 0$ (δηλαδή όταν η εξίσωση είναι πρωτοβάθμια).

5.1.1.1 Πολυωνυμικές εξισώσεις

Με τις συναρτήσεις **Solve** και **Reduce** μπορούμε να επιλύσουμε πολλά διαφορετικά είδη εξισώσεων, αλλά τόσο αυτές οι δύο όσο και η συνάρτηση **Roots**, που θα γνωρίσουμε παρακάτω, είναι ιδιαίτερα κατάλληλες για την επίλυση πολυωνυμικών εξισώσεων.

Οι συναρτήσεις **Solve** και **Reduce** επιλύνουν με ακρίβεια κάθε πολυωνυμική εξίσωση, η οποία είναι βαθμού $n \leq 4$. Όταν, όμως, είναι $n \geq 5$, τότε δεν τις επιλύνουν πάντα, χωρίς βέβαια να αποκλείεται κάτι τέτοιο. Στην περίπτωση αυτή το *Mathematica* χρησιμοποιεί εκφράσεις της μορφής Root για να τις αναπαραστήσει. Όταν συμβεί κάτι τέτοιο, μπορούμε να πάρουμε εκ των υστέρων αριθμητικές τιμές για τις λύσεις με τη συνάρτηση **N**.

Παράδειγμα 3: Να λυθεί η εξίσωση $x^4 + 2x^3 - 13x^2 - 14x + 24 = 0$.

Επιλύνουμε πρώτα την εξίσωση με τη συνάρτηση **Solve**:

```
Solve[x^4 + 2 x^3 - 13 x^2 - 14 x + 24 == 0, x]

{{x → -4}, {x → -2}, {x → 1}, {x → 3}}
```

και στη συνέχεια με τη συνάρτηση **Reduce**:

```
Reduce[x^4 + 2 x^3 + -13 x^2 - 14 x + 24 == 0, x]

x == -4 || x == -2 || x == 1 || x == 3
```

Μετατροπή των λύσεων σε κανόνες αντικατάστασης:

```
{ToRules[%]}

{{x → -4}, {x → -2}, {x → 1}, {x → 3}}
```

Παρατηρούμε, ότι και οι δύο επέστρεψαν τις ίδιες λύσεις.

Παράδειγμα 4: Να λυθεί η εξίσωση $x^5 + 4x - 1 = 0$.

Επιλύνουμε πρώτα την εξίσωση με τη συνάρτηση **Solve**:

```
Solve[x^5 + 4 x - 1 == 0, x]

{{x → Root[-1 + 4 #1 + #1^5 &, 1]},
 {x → Root[-1 + 4 #1 + #1^5 &, 2]}, {x → Root[-1 + 4 #1 + #1^5 &, 3]},
 {x → Root[-1 + 4 #1 + #1^5 &, 4]}, {x → Root[-1 + 4 #1 + #1^5 &, 5]}}
```

και στη συνέχεια με τη συνάρτηση **Reduce**:

```
Reduce[x^5 + 4 x - 1 == 0, x]

Root[-1 + 4 #1 + #1^5 &, 1] == x ||
 Root[-1 + 4 #1 + #1^5 &, 2] == x || Root[-1 + 4 #1 + #1^5 &, 3] == x ||
 Root[-1 + 4 #1 + #1^5 &, 4] == x || Root[-1 + 4 #1 + #1^5 &, 5] == x
```

Παρατηρούμε ότι το *Mathematica* δεν μπορεί να βρει τις ακριβείς λύσεις της εξίσωσης ούτε με τη συνάρτηση **Solve** ούτε με τη συνάρτηση **Reduce**. Μπορούμε όμως να πάρουμε αριθμητικές (προσεγγιστικές) λύσεις της εξίσωσης με τη συνάρτηση **N**:

```
N[%%]

{{x → 0.249757}, {x → -1.05775 - 1.00384 i},
 {x → -1.05775 + 1.00384 i},
 {x → 0.932871 - 1.00627 i}, {x → 0.932871 + 1.00627 i}}
```

Παράδειγμα 5: Να λυθεί η εξίσωση $x^7 - x^5 + 4x^3 - x^2 - 4x + 1 = 0$.

Επιλύνουμε την εξίσωση με τη συνάρτηση **Solve**:

```
Solve[x^7 - x^5 + 4 x^3 - x^2 - 4 x + 1 == 0, x]

{{x → -1}, {x → 1}, {x → Root[-1 + 4 #1 + #1^5 &, 1]},
 {x → Root[-1 + 4 #1 + #1^5 &, 2]}, {x → Root[-1 + 4 #1 + #1^5 &, 3]},
 {x → Root[-1 + 4 #1 + #1^5 &, 4]}, {x → Root[-1 + 4 #1 + #1^5 &, 5]}}
```

Παρατηρούμε ότι το *Mathematica* επιστρέφει δύο ακριβείς λύσεις, τις $x_0 = -1$ και $x_1 = 1$ ενώ τις υπόλοιπες τις αναπαριστά με χρήση εκφράσεων της μορφής Root. Αυτό είναι αναμενόμενο, γιατί αν αναλύσουμε το πολυώνυμο σε γινόμενο πολυωνύμων μικρότερου βαθμού, με τη χρήση της συνάρτησης **Factor,** θα έχουμε το εξής αποτέλεσμα:

```
Factor[x^7 - x^5 + 4 x^3 - x^2 - 4 x + 1]

(-1 + x) (1 + x) (-1 + 4 x + x^5)
```

Εύρεση αριθμητικών (προσεγγιστικών) λύσεων της εξίσωσης:

```
N[%%]

{{x → -1.}, {x → 1.}, {x → 0.249757},
 {x → -1.05775 - 1.00384 i}, {x → -1.05775 + 1.00384 i},
 {x → 0.932871 - 1.00627 i}, {x → 0.932871 + 1.00627 i}}
```

Εκτός από τις συναρτήσεις **Solve** και **Reduce**, μια άλλη συνάρτηση, που μπορούμε να χρησιμοποιήσουμε για να λύνουμε εξισώσεις, είναι η **Roots**, η σύνταξη της οποίας είναι παρόμοια με των άλλων δύο. Επιλύνει όμως **μόνον πολυωνυμικές εξισώσεις** και δίνει λύσεις της μορφής:

$$x == x_0 \,||\, x == x_1 \ldots.$$

Παράδειγμα 6: Να λυθεί η εξίσωση $a x^2 - 2x + 4 = 0$.

Επίλυση της εξίσωσης με τη συνάρτηση **Roots**:

Roots[a x^2 - 2 x + 4 == 0, x]

$$x == \frac{2-\sqrt{4-16a}}{2a} \;||\; x == \frac{2+\sqrt{4-16a}}{2a}$$

Μετατροπή των λύσεων σε κανόνες αντικατάστασης:

{ToRules[%]}

$$\left\{\left\{x \to \frac{2-\sqrt{4-16a}}{2a}\right\}, \left\{x \to \frac{2+\sqrt{4-16a}}{2a}\right\}\right\}$$

Παρατήρηση: Οι εντολές **Solve**[eqn,x] και {**ToRules**[**Roots**[eqn,x]]}, επιστρέφουν το ίδιο αποτέλεσμα.

Solve[a x^2 - 2 x + 4 == 0, x]

$$\left\{\left\{x \to \frac{2-\sqrt{4-16a}}{2a}\right\}, \left\{x \to \frac{2+\sqrt{4-16a}}{2a}\right\}\right\}$$

{ToRules[Roots[a x^2 - 2 x + 4 == 0, x]]}

$$\left\{\left\{x \to \frac{2-\sqrt{4-16a}}{2a}\right\}, \left\{x \to \frac{2+\sqrt{4-16a}}{2a}\right\}\right\}$$

Με τις συναρτήσεις **Solve** και **Reduce** μπορούμε να λύσουμε και εξισώσεις που περιλαμβάνουν είτε ριζικά είτε κλάσματα, όπως επίσης και τριγωνομετρικές εξισώσεις, εκθετικές και λογαριθμικές εξισώσεις.

5.1.1.2 Εξισώσεις με ριζικά

Παράδειγμα 7: Να λυθεί η εξίσωση $\frac{1}{x} - \frac{1}{\sqrt{x}} = 2$.

Επίλυση της εξίσωσης με τη συνάρτηση **Solve**:

```
Solve[1/x - 1/Sqrt[x] == 2, x]
```

$\{\{x \to \frac{1}{4}\}\}$

Επίλυση της εξίσωσης με τη συνάρτηση **Reduce**:

```
Reduce[1/x - 1/Sqrt[x] == 2, x]
```

$x == \frac{1}{4}\ \&\&\ x \neq 0$

Εύρεση αριθμητικών (προσεγγιστικών) λύσεων της εξίσωσης:

```
N[%%]
```

$\{\{x \to 0.25\}\}$

5.1.1.3 Κλασματικές εξισώσεις

Παράδειγμα 8: Να λυθεί η εξίσωση $\frac{x}{x^2-1} - \frac{1}{x+1} = \frac{x^2}{x^2+1}$.

Επίλυση της εξίσωσης με τη συνάρτηση **Solve**:

```
Solve[x/(x^2 - 1) - 1/(x + 1) == x^2/(x^2 + 1), x]
```

$\{\{x \to -i\sqrt{-1+\sqrt{2}}\}, \{x \to i\sqrt{-1+\sqrt{2}}\}, \{x \to -\sqrt{1+\sqrt{2}}\}, \{x \to \sqrt{1+\sqrt{2}}\}\}$

Επίλυση της εξίσωσης με τη συνάρτηση **Reduce**:

```
Reduce[x/(x^2 - 1) - 1/(x + 1) == x^2/(x^2 + 1), x]
```

$x == -i\sqrt{-1+\sqrt{2}} \;||\; x == i\sqrt{-1+\sqrt{2}} \;||\; x == -\sqrt{1+\sqrt{2}} \;||\; x == \sqrt{1+\sqrt{2}}$

Εύρεση αριθμητικών (προσεγγιστικών) λύσεων της εξίσωσης:

```
N[%%]
```

```
{{x → 0. - 0.643594 i},
 {x → 0. + 0.643594 i}, {x → -1.55377}, {x → 1.55377}}
```

5.1.1.4 Τριγωνομετρικές εξισώσεις

Παράδειγμα 9: Να λυθεί η εξίσωση Sin[x] = $\frac{1}{2}$.

Είναι γνωστό ότι η εξίσωση Sin[x] = $\frac{1}{2}$ έχει άπειρες λύσεις της μορφής $x = \pi/6 + 2k\pi$, $k \in \mathbb{R}$. Αν επιχειρήσουμε να λύσουμε αυτή την εξίσωση με κάποια από τις συναρτήσεις **Solve** ή **Reduce**, θα πάρουμε αρχικά ένα μήνυμα, με το οποίο το πρόγραμμα μας προειδοποιεί ότι έχουν χρησιμοποιηθεί αντίστροφες συναρτήσεις κατά την επίλυση με συνέπεια κάποιες λύσεις μιθανόν να μην βρεθούν, και στη συνέχεια το πρόγραμμα παρουσιάζει μόνο μία λύση (αυτή που προκύπτει για k = 0).

Επίλυση της εξίσωσης με τη συνάρτηση **Solve**:

```
Solve[Sin[x] == 1/2, x]
```

```
Solve::ifun :
 Inverse functions are being used by Solve, so some solutions may not be found.
```

$\{\{x \to \frac{\pi}{6}\}\}$

Επίλυση της εξίσωσης με τη συνάρτηση **Reduce** (για την έκδοση ***Mathematica*** 4):

```
Reduce[Sin[x] == 1/2, x]
```

```
Reduce::ifun :
 Inverse functions are being used by Reduce, so some solutions may not be found.
```

$x == \frac{\pi}{6}$

Οι συναρτήσεις **Solve** ή **Reduce**(η Reduce στο *Mathematica* 5 δεν έχει αυτή την επιλογή και επίσης λύνει άνετα την παραπάνω εξίσωση), έχουν μία επιλογή την **InverseFunctions**, η οποία καθορίζει κατά πόσο αντίστροφες συναρτήσεις μπορούν να χρησιμοποιηθούν κατά την επίλυση.

Οι ρυθμίσεις της επιλογής **InverseFunctions** είναι οι εξής:

True	αντίστροφες συναρτήσεις θα χρησιμοποιούνται πάντα.
Automatic	αντίστροφες συναρτήσεις θα χρησιμοποιούνται, αλλά θα εκτυπώνεται ένα μήνυμα (default).
False	αντίστροφες συναρτήσεις δεν θα χρησιμοποιούνται.

Όταν δεν χρησιμοποιούμε αυτή την επιλογή στις συναρτήσεις **Solve** ή **Reduce**, θεωρείται ότι η επιλογή **InverseFunctions** έχει τη ρύθμιση Automatic.

Επίλυση της εξίσωσης με τη συνάρτηση **Solve** και με χρήση της επλογής **InverseFunctios**:

```
Solve[Sin[x] == 1/2, x, InverseFunctions → True]
```

$\{\{x \to \frac{\pi}{6}\}\}$

Επίλυση της εξίσωσης με τη συνάρτηση **Reduce**(στο *Mathematica* 4) με χρήση της επιλογής **InverseFunctios**:

```
Reduce[Sin[x] == 1/2, x, InverseFunctions → True]
```

$x == \frac{\pi}{6}$

Εύρεση αριθμητικής (προσεγγιστικής) λύσης της εξίσωσης:

```
N[%%]
```

{{x → 0.523599}}

5.1.1.5 Εκθετικές και λογαριθμικές εξισώσεις

Παράδειγμα 10: Να λυθεί η εξίσωση $(2\,e)^x + 4\,e = 4\,e^x + e\,2^x$.

Επίλυση της εξίσωσης με τη συνάρτηση **Solve**:

```
Solve[(2 e)^x + 4 e == 4 e^x + e 2^x, x]
```

Solve::ifun :
Inverse functions are being used by Solve, so some solutions may not be found.

{{x → 1}, {x → 2}}

Επίλυση της εξίσωσης στο *Mathematica* 4 με τη συνάρτηση **Reduce** και με χρήση της επιλογής **InverseFunctios**:

```
Reduce[(2 e)^x + 4 e == 4 e^x + e 2^x, x, InverseFunctions → True]

x == 1 || x == 2
```

Παράδειγμα 11: Να λυθεί η εξίσωση $\ln\left(\sqrt{x}\right) = \sqrt{\ln(x)}$.

Επίλυση της εξίσωσης με τη συνάρτηση **Solve**:

```
Solve[Log[Sqrt[x]] == Sqrt[Log[x]], x]

{{x → 1}, {x → e^4}}
```

Επίλυση της εξίσωσης με τη συνάρτηση **Reduce**:

```
Reduce[Log[Sqrt[x]] == Sqrt[Log[x]], x]

x == 1 || x == e^4
```

5.1.2 Αριθμητική (προσεγγιστική) επίλυση εξισώσεων

Δύο είναι οι βασικές συναρτήσεις με τις οποίες μπορούμε να πάρουμε αριθμητικές (προσεγγιστικές) λύσεις εξισώσεων. Η μία είναι η συνάρτηση **NSolve**[eqn, x], η οποία εφαρμόζεται κυρίως σε πολυωνυμικές και γενικά σε εξισώσεις, που μπορεί να επιλύσει η συνάρτηση **Solve**. Η άλλη είναι η συνάρτηση **FindRoot**[eqn, {x, x_0}], η οποία αναζητεί μια αριθμητική λύση της εξίσωσης στην περιοχή του x = x_0.

Παράδειγμα 12: Να λυθεί η εξίσωση $x^3 + 4x - 1 = 0$.

Για να βρούμε τις αριθμητικές λύσεις της εξίσωσης, μπορούμε να τη λύσουμε πρώτα με τη συνάρτηση **Solve**, για να πάρουμε τις ακριβείς λύσεις της (κάτι που είναι εφικτό αφού η εξίσωση είναι τρίτου βαθμού),

```
Solve[x^3 + 4 x - 1 == 0, x]
```

$$\left\{\left\{x \to -4\left(\frac{2}{3\left(9+\sqrt{849}\right)}\right)^{1/3} + \frac{\left(\frac{1}{2}\left(9+\sqrt{849}\right)\right)^{1/3}}{3^{2/3}}\right\},\right.$$
$$\left\{x \to 2\left(1+i\sqrt{3}\right)\left(\frac{2}{3\left(9+\sqrt{849}\right)}\right)^{1/3} - \frac{\left(1-i\sqrt{3}\right)\left(\frac{1}{2}\left(9+\sqrt{849}\right)\right)^{1/3}}{2\ 3^{2/3}}\right\},$$
$$\left.\left\{x \to 2\left(1-i\sqrt{3}\right)\left(\frac{2}{3\left(9+\sqrt{849}\right)}\right)^{1/3} - \frac{\left(1+i\sqrt{3}\right)\left(\frac{1}{2}\left(9+\sqrt{849}\right)\right)^{1/3}}{2\ 3^{2/3}}\right\}\right\}$$

και μετά να μετατρέψουμε τις ακριβείς λύσεις σε αριθμητικές με τη συνάρτηση **N**:

```
N[%]

{{x → 0.246266}, {x → -0.123133 + 2.01134 ⅈ},
 {x → -0.123133 - 2.01134 ⅈ}}
```

Αν δεν μας ενδιαφέρουν οι ακριβείς λύσεις, τότε εφαμόζουμε απ' ευθείας τη συνάρτηση **NSolve**:

```
NSolve[x^3 + 4 x - 1 == 0, x]

{{x → -0.123133 - 2.01134 ⅈ},
 {x → -0.123133 + 2.01134 ⅈ}, {x → 0.246266}}
```

Με τη συνάρτηση **NSolve** μπορούμε να επιλύσουμε αριθμητικά σχεδόν όλα τα είδη εξισώσεων. Για παράδειγμα:

Παράδειγμα 13: Να λυθεί αριθμητικά η εξίσωση $2x^2 - \sqrt{x} = 2$.

```
NSolve[2 x^2 - Sqrt[x] == 2, x]

{{x → 1.24847}}
```

Παράδειγμα 14: Να λυθεί αριθμητικά η εξίσωση $\frac{1}{x+1} + \frac{1}{x^2-1} = 2$.

```
NSolve[1/(x + 1) + 1/(x^2 - 1) == 2, x]

{{x → 1.28078}, {x → -0.780776}}
```

Παράδειγμα 15: Να λυθεί αριθμητικά η εξίσωση $\sin x + 2\cos x = 1$.

```
NSolve[Sin[x] + 2 Cos[x] == 1, x]

Solve::ifun :
 Inverse functions are being used by Solve, so some solutions may not be found.

{{x → -0.643501}, {x → 1.5708}}
```

Παράδειγμα 16: Να λυθεί αριθμητικά η εξίσωση $\ln\left(x^2 - \sqrt{1-x}\right) = 1$.

```
NSolve[Log[x^2 - √(1 - x)] == 1, x]

{{x → -2.11753}}
```

Η εντολή **NSolve** δεν βρίσκει αριθμητικές λύσεις για όλα τα είδη των εξισώσεων.

Παράδειγμα 17: Να λυθεί αριθμητικά η εξίσωση $\cos x - x^2 = 0$.

Παρατηρούμε ότι η εξίσωση δεν μπορεί να λυθεί αριθμητικά με τη χρήση της συνάρτησης **NSolve**.

```
NSolve[Cos[x] - x^2 == 0, x]

Solve::tdep : The equations appear to involve the
   variables to be solved for in an essentially non-algebraic way.

NSolve[-x^2 + Cos[x] == 0, x]
```

Σε τέτοιες περιπτώσεις, και όχι μόνο σε αυτές, χρησιμοποιούμε τη δεύτερη συνάρτηση που διαθέτει το *Mathematica,* τη συνάρτηση **FindRoot**[eqn, {x, x_0}], η οποία αναζητεί μια αριθμητική λύση της εξίσωσης στην περιοχή του x = x_0. Συγκεκριμένα, αναζητούμε μία αριθμητική λύση στη περιοχή του x = 1.

```
FindRoot[Cos[x] - x^2 == 0, {x, 1}]

{x → 0.824132}
```

Μια καλή τακτική, που μπορούμε να ακολουθήσουμε σε συνδυασμό με τη συνάρτηση **FindRoot**, είναι να κάνουμε πρώτα τη γραφική παράσταση της συνάρτησης, της οποία ο μηδενισμός μας δίνει της δίνει την προ επίλυση εξίσωση, με τη συνάρτηση:

Plot[f(x), {x, xminx, xmax}]

(στην οποία θα αναφερθούμε αναλυτικά σε επόμενο κεφάλαιο), σε ένα κατάλληλο διάστημα [xmin, xmax]. Από τη γραφική παράσταση λαμβάνουμε κατά προσέγγιση τα σημεία τομής της με τον άξονα των x και τα χρησιμοποιούμε ως σημεία αφετηρίας στη συνάρτηση **FindRoot** (δηλαδή, αναζητούμε λύσεις στην περιοχή αυτών των σημείων).

Συγκεκριμένα, στο παράδειγμα μας, κάνουμε τη γραφική παράσταση της συνάρτησης f(x) = cos(x) - x^2 στο διάστημα [-π/3, π/3] με τη χρήση της συνάρτησης **Plot**:

```
Plot[Cos[x] - x^2, {x, -π / 3, π / 3}, PlotStyle → RGBColor[1, 0, 0]]
```




```
- Graphics -
```

στην οποία χρησιμοποιήσαμε την επιλογή **PlotStyle**→RGBColor[1,0,0] για να εμφανίσουμε τη γραμμή της καμπύλης με κόκκινο χρώμα. Από τη γραφική παράσταση πο προέκυψε, βρίσκουμε ότι τα σημεία τομής της με το άξονα των χ είναι τα σημεία με τετμημένη $x \cong -0{,}8$ και $x \cong 0{,}8$. Στη συνέχεια με αφετηρία της συνάρτησης **FindRoot** αυτές τις δύο τιμές βρίσκουμε τις προσεγγιστικές λύσεις της εξίσωσης:

```
FindRoot[Cos[x] - x^2 == 0, {x, 0.8}]

{x → 0.824132}
```

```
FindRoot[Cos[x] - x^2 == 0, {x, -0.8}]

{x → -0.824132}
```

5.2 Επίλυση συστημάτων

Οι βασικές συναρτήσεις που χρησιμοποιούμε για την επίλυση συστημάτων είναι οι **Solve** και **Reduce**, τις οποίες συντάσουμε κατά τα γνωστά, με μόνη διαφορά, ότι τις εξισώσεις του συστήματος είτε τις γράφουμε με μορφή λίστας είτε τις συνδέουμε με το λογικό τελεστή && (και) ενώ τους αγνώστους τους δηλώνουμε επίσης με μορφή λίστας {x, y, ...}.

Οι λύσεις $(x_0, y_0, \ldots), (x_1, y_1, \ldots), \ldots$ παρουσιάζονται:

α) Σε λίστα υπό τη μορφή κανόνων αντικατάστασης:

$$\{\{x \to x_0,\ y \to y_0, \ldots\}, \{x \to x_1,\ y \to y_1, \ldots\}, \ldots\}$$

όταν χρησιμοποιούμε τη **Solve**. Για να τις εμφανίσουμε σε μορφή λίστας, χωρίς τους κανόνες αντικατάστασης,
μπορούμε κατά τα γνωστά να χρησιμοποιήσουμε τον τελεστή αντικταστάσης "/.".

β) Υπό τη μορφή:

$$x == x_0\ \&\&\ y == y_0\ \&\&\ \ldots\ ||\ x == x_1\ \&\&\ y == y_1\ \&\&\ \ldots\ ||\ \ldots$$

όταν χρησιμοποιούμε τη **Reduce**. Για να πάρουμε τις λύσεις σε μορφή λίστας εφαρμόζουμε πρώτα την εντολή {**ToRules**[%]} για να τις εμφανίσουμε ως κανόνες αντικατάστασης και στη συνέχεια εφαρμόζουμε
τον τελεστή αντικτάστασης {x, y, ...} /. %. Ισοδύναμα μπορούμε να εφαμόσουμε το σύνθετο τελεστή {x, y, ...} /. {**ToRules**[%]}.

Παράδειγμα 18: Έστω το σύστημα των δύο γραμμικών εξισώσεων

$$a\,x + b\,y = c$$
$$d\,x + e\,y = f$$

με δύο αγνώστους x και y. Επιλύνοντάς το με τη συνάρτηση **Solve** βρίσκουμε τις ακριβείς λύσεις του συστήματος

```
Solve[{a x + b y == c, d x + e y == f}, {x, y}]
```

$$\left\{\left\{x \to -\frac{-c\,e + b\,f}{-b\,d + a\,e},\ y \to -\frac{-c\,d + a\,f}{b\,d - a\,e}\right\}\right\}$$

ενώ επιλύνοντάς το με τη συνάρτηση **Reduce** παίρνουμε μια πλήρη διερεύνηση του συστήματος

```
Reduce[a x + b y == c && d x + e y == f, {x, y}]
```

$x == \frac{-c e + b f}{b d - a e} \;\&\&\; y == \frac{c d - a f}{b d - a e} \;\&\&\; b d - a e \neq 0 \;||$
$b == \frac{a e}{d} \;\&\&\; c == \frac{a f}{d} \;\&\&\; x == \frac{f - e y}{d} \;\&\&\; d \neq 0 \;||$
$a == 0 \;\&\&\; c == \frac{b f}{e} \;\&\&\; d == 0 \;\&\&\; y == \frac{f}{e} \;\&\&\; e \neq 0 \;||$
$d == 0 \;\&\&\; e == 0 \;\&\&\; f == 0 \;\&\&\; x == \frac{c - b y}{a} \;\&\&\; a \neq 0 \;||$
$a == 0 \;\&\&\; b == 0 \;\&\&\; c == 0 \;\&\&\; d == 0 \;\&\&\; e == 0 \;\&\&\; f == 0 \;||$
$a == 0 \;\&\&\; d == 0 \;\&\&\; e == 0 \;\&\&\; f == 0 \;\&\&\; y == \frac{c}{b} \;\&\&\; b \neq 0$

5.2.1 Συστήματα με μία παράμετρο

Παράδειγμα 19: Έστω το σύστημα

$$k x^2 + y = 1$$
$$x + y = 1$$

όπου τα x και y είναι οι άγνωστοι και k παράμετρος του συστήματος.

Αρχικά δίνουμε ένα όνομα (sys1) στο σύστημα για να μπορέσουμε στη συνέχεια να αναφερόμαστε σε αυτό.

```
sys1 = {k x^2 + y == 1, x + y == 1}
```

$\{k x^2 + y == 1, x + y == 1\}$

Στη συνέχεια επιλύνουμε το σύστημα (με τη χρήση της συνάρτησης **Solve**) και τη λύση την αποθηκεύουμε στην μεταβλητή sol1.

```
sol1 = Solve[sys1, {x, y}]
```

$\left\{\{y \to 1, x \to 0\}, \left\{y \to \frac{-1 + k}{k}, x \to \frac{1}{k}\right\}\right\}$

Μπορούμε να κάνουμε επαλήθευση χρησιμοποιοώντας τον τελεστή "/."

```
sys1 /. sol1
```

$\left\{\{\text{True}, \text{True}\}, \left\{\frac{1}{k} + \frac{-1 + k}{k} == 1, \frac{1}{k} + \frac{-1 + k}{k} == 1\right\}\right\}$

Παρατηρούμε ότι για το δεύτερο ζεύγος λύσεων η επαλήθευση δεν είναι σαφής. Στην περίπτωση αυτή μπορούμε να χρησιμοποιήσουμε τη συνάρτηση **Simplify**:

```
sys1 /. sol1 // Simplify

{{True, True}, {True, True}}
```

Τέλος, μπορούμε να κάνουμε πλήρη διερεύνηση για τις διάφορες τιμές της παραμέτρου k με τη συνάρτηση **Reduce**:

```
Reduce[sys1, {x, y}]
```

$x == 0 \&\& y == 1 \,||\, x == \frac{1}{k} \&\& y == \frac{-1+k}{k} \&\& k \neq 0$

5.2.2 Συστήματα με δύο παραμέτρους

Παράδειγμα 20: Έστω το σύστημα

$$a x + b y = 1$$
$$x - b y = 2$$

όπου τα x και y είναι οι άγνωστοι και τα a, b οι παράμετροι του συστήματος.

Επιλύουμε το σύστημα χρησιμοποιώντας τη συνάρτηση **Solve**:

```
Solve[{a x + b y == 1, x - b y == 2}, {x, y}]
```

$\left\{\left\{x \to \frac{3}{1+a},\ y \to -\frac{-1+2a}{(1+a)\,b}\right\}\right\}$

Κάνουμε επαλήθευση χρησιμοποιοώντας τον τελεστή "/."

```
{a x + b y == 1, x - b y == 2} /. %
```

$\left\{\left\{\frac{3a}{1+a} - \frac{-1+2a}{1+a} == 1,\ \frac{3}{1+a} + \frac{-1+2a}{1+a} == 2\right\}\right\}$

Επειδή η επαλήθευση δεν είναι σαφής χρησιμοποιήσουμε τη συνάρτηση **Simplify**:

```
{a x + b y == 1, x - b y == 2} /. %% // Simplify

{{True, True}}
```

Τέλος, κάνουμε πλήρη διερεύνηση για τις διάφορες τιμές των παραμέτρων a, b με τη συνάρτηση **Reduce**:

```
Reduce[{a x + b y == 1, x - b y == 2}, {x, y}]
```

$a == \frac{1}{2}$ && $b == 0$ && $x == 2$ ||
$x == \frac{3}{1+a}$ && $y == \frac{1-2a}{(1+a)\,b}$ && $1 + a \neq 0$ && $b \neq 0$

5.2.3 Συστήματα βαθμού μεγάλυτερου του 2

Για τα συστήματα βαθμού $n \geq 2$ ισχύουν τα ίδια ότι και για τα συστήματα πρώτου βαθμού.

Παράδειγμα 21: Έστω το σύστημα:

$$x^2 - y = 3$$
$$x - y = 2$$

Μπορούμε να βρούμε τις ακριβείς λύσεις του χρησιμοποιώντας τη συνάρτηση **Solve**:

```
Solve[{x^2 - y == 3, x - y == 2}, {x, y}]
```

$\left\{\left\{y \to -\frac{3}{2} - \frac{\sqrt{5}}{2},\ x \to \frac{1}{2}\left(1-\sqrt{5}\right)\right\},\ \left\{y \to \frac{1}{2}\left(-3+\sqrt{5}\right),\ x \to \frac{1}{2}\left(1+\sqrt{5}\right)\right\}\right\}$

και να τις μετατρέψουμε σε αριθμητικές (προσεγγιστικές) λύσεις με τη συνάρτηση **N**:

```
N[%]
```

$\{\{y \to -2.61803,\ x \to -0.618034\},\ \{y \to -0.381966,\ x \to 1.61803\}\}$

Επίσης τις αριθμητικές (προσεγγιστικές) λύσεις του συστήματος, μπορούμε και να τις βρούμε απευθείας χρησιμοποιώντας τη συνάρτηση **NSolve**:

```
NSolve[{x^2 - y == 3, x - y == 2}, {x, y}]
```

$\{\{y \to -0.381966,\ x \to 1.61803\},\ \{y \to -2.61803,\ x \to -0.618034\}\}$

5.2.4 Συστήματα με περισσότερους αγνώστους

Παράδειγμα 22: Έστω το παρακάτω σύστημα τριών εξισώσεων, με αγνώστους x, y, z:

$$x + 2y + z = 16$$
$$x - 2y + z = 0$$
$$2x + y - z = 5$$

Επιλύνουμε το σύστημα χρησιμοποιώντας τη συνάρτηση **Solve**:

```
Solve[{x + 2 y + z == 16, x - 2 y + z == 0, 2 x + y - z == 5}, {x, y, z}]
```

$\{\{x \to 3,\ y \to 4,\ z \to 5\}\}$

Κάνουμε επαλήθευσηχρησιμοποιοώντας τον τελεστή "/."

```
{x + 2 y + z == 16, x - 2 y + z == 0, 2 x + y - z == 5} /. %
```

{{True, True, True}}

Παράδειγμα 23: Έστω το σύστημα

$$\alpha x + y + z = 1$$
$$x + \alpha y + z = \alpha$$
$$x + y + \alpha z = a^2$$

όπου τα x, y, z είναι οι άγνωστοι και α παράμετρος του συστήματος.

Αρχικά δίνουμε ένα όνομα (sys2) στο σύστημα για να μπορέσουμε στη συνέχεια να αναφερόμαστε σε αυτό.

```
sys2 = {a x + y + z == 1, x + a y + z == a, x + y + a z == a^2}
```

$\{a x + y + z == 1,\ x + a y + z == a,\ x + y + a z == a^2\}$

Επιλύνουμε το σύστημα (με τη χρήση της συνάρτησης **Solve**) και τη λύση την αποθηκεύουμε στην μεταβλητή sol2.

```
sol2 = Solve[sys2, {x, y, z}]
```

$$\left\{\left\{x \to -\frac{1+a}{2+a},\ y \to \frac{1}{2+a},\ z \to -\frac{-1-2a-a^2}{2+a}\right\}\right\}$$

Κάνουμε επαλήθευση χρησιμοποιοώντας τον τελεστή "/."

```
sys2 /. sol2
```

$$\left\{\left\{\frac{1}{2+a} - \frac{a(1+a)}{2+a} - \frac{-1-2a-a^2}{2+a} == 1,\right.\right.$$
$$\frac{a}{2+a} - \frac{1+a}{2+a} - \frac{-1-2a-a^2}{2+a} == a,$$
$$\left.\left.\frac{1}{2+a} - \frac{1+a}{2+a} - \frac{a(-1-2a-a^2)}{2+a} == a^2\right\}\right\}$$

Επειδή η επαλήθευση δεν είναι σαφής χρησιμοποιήσουμε τη συνάρτηση **Simplify**:

```
sys2 /. sol2 // Simplify

{{True, True, True}}
```

Τέλος, κάνουμε πλήρη διερεύνηση για τις διάφορες τιμές της παραμέτρου α με τη συνάρτηση **Reduce**:

```
Reduce[sys2, {x, y, z}]
```

$$a == 1 \;\&\&\; x == 1 - y - z \;\|\; x == \frac{-1 - a}{2 + a} \;\&\&\; y == \frac{1}{2 + a} \;\&\&\; z == \frac{1 + 2 a + a^2}{2 + a} \;\&\&\; -1 + a \neq 0 \;\&\&\; 2 + a \neq 0$$

5.2.5 Εύρεση προσεγγιστικών λύσεων

Φυσικά υπάρχουν συστήματα, τα οποία δεν μπορούν να επιλυθούν με ακρίβεια.

Παράδειγμα 24: Έστω το σύστημα

$$x^2 + \frac{1}{y^2} = 1$$
$$\frac{1}{x^4} - \frac{1}{y^6} = 2$$

Αρχικά δίνουμε ένα όνομα (sys3) στο σύστημα για να μπορέσουμε στη συνέχεια να αναφερόμαστε σε αυτό:

$$\mathbf{sys3 = \left\{x^2 + \frac{1}{y^2} == 1,\ \frac{1}{x^4} - \frac{1}{y^6} == 2\right\}}$$

$$\left\{x^2 + \frac{1}{y^2} == 1,\ \frac{1}{x^4} - \frac{1}{y^6} == 2\right\}$$

Επιλύουμε το σύστημα (με τη χρήση της συνάρτησης **Solve**) και τη λύση την αποθηκεύουμε στην μεταβλητή sol3:

```
sol3 = Solve[sys3, {x, y}]
```

$$\{\{x \to -\sqrt{(-1 + \text{Root}[1 - 2\,\#1 + \#1^2 + 2\,\#1^3 - 4\,\#1^4 + \#1^5\,\&, 1] + 2\,\text{Root}[1 - 2\,\#1 + \#1^2 + 2\,\#1^3 - 4\,\#1^4 + \#1^5\,\&, 1]^2 - 4\,\text{Root}[1 - 2\,\#1 + \#1^2 + 2\,\#1^3 - 4\,\#1^4 + \#1^5\,\&, 1]^3 + \text{Root}[1 - 2\,\#1 + \#1^2 + 2\,\#1^3 - 4\,\#1^4 + \#1^5\,\&, 1]^4)},$$
$$y \to -\sqrt{\text{Root}[1 - 2\,\#1 + \#1^2 + 2\,\#1^3 - 4\,\#1^4 + \#1^5\,\&, 1]}\},$$
$$\{x \to -\sqrt{(-1 + \text{Root}[1 - 2\,\#1 + \#1^2 + 2\,\#1^3 - 4\,\#1^4 + \#1^5\,\&, 1] + 2\,\text{Root}[1 - 2\,\#1 + \#1^2 + 2\,\#1^3 - 4\,\#1^4 + \#1^5\,\&, 1]^2 - 4\,\text{Root}[1 - 2\,\#1 + \#1^2 + 2\,\#1^3 - 4\,\#1^4 + \#1^5\,\&, 1]^3 +$$

```
      Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1]^4),
   y → √(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1])},
  {x → √(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1] +
      2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1]^2 -
      4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1]^3 +
      Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1]^4),
   y → -√(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1])},
  {x → √(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1] +
      2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1]^2 -
      4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1]^3 +
      Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1]^4),
   y → √(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 1])},
  {x → -√(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2] +
      2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^2 -
      4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^3 +
      Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^4),
   y → -√(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2])},
  {x → -√(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2] +
      2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^2 -
      4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^3 +
      Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^4),
   y → √(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2])},
  {x → √(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2] +
      2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^2 -
      4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^3 +
      Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^4),
   y → -√(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2])},
  {x → √(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2] +
      2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^2 -
      4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^3 +
      Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2]^4),
   y → √(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 2])},
  {x → -√(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 3] +
      2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 3]^2 -
      4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 3]^3 +
      Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 3]^4),
   y → -√(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 3])},
```

$$\left\{x\to -\sqrt{\left(-1+\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]+\right.}\right.$$
$$2\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]^2-$$
$$4\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]^3+$$
$$\left.\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]^4\right),$$
$$\left.y\to\sqrt{\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]}\right\},$$
$$\left\{x\to \sqrt{\left(-1+\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]+\right.}\right.$$
$$2\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]^2-$$
$$4\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]^3+$$
$$\left.\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]^4\right),$$
$$\left.y\to-\sqrt{\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]}\right\},$$
$$\left\{x\to \sqrt{\left(-1+\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]+\right.}\right.$$
$$2\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]^2-$$
$$4\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]^3+$$
$$\left.\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]^4\right),$$
$$\left.y\to\sqrt{\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,3\right]}\right\},$$
$$\left\{x\to -\sqrt{\left(-1+\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]+\right.}\right.$$
$$2\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^2-$$
$$4\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^3+$$
$$\left.\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^4\right),$$
$$\left.y\to-\sqrt{\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]}\right\},$$
$$\left\{x\to -\sqrt{\left(-1+\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]+\right.}\right.$$
$$2\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^2-$$
$$4\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^3+$$
$$\left.\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^4\right),$$
$$\left.y\to\sqrt{\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]}\right\},$$
$$\left\{x\to \sqrt{\left(-1+\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]+\right.}\right.$$
$$2\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^2-$$
$$4\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^3+$$
$$\left.\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^4\right),$$
$$\left.y\to-\sqrt{\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]}\right\},$$
$$\left\{x\to \sqrt{\left(-1+\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]+\right.}\right.$$
$$2\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^2-$$
$$4\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^3+$$
$$\left.\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]^4\right),$$
$$\left.y\to\sqrt{\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,4\right]}\right\},$$
$$\left\{x\to -\sqrt{\left(-1+\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,5\right]+\right.}\right.$$
$$2\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,5\right]^2-$$
$$4\,\text{Root}\left[1-2\,\#1+\#1^2+2\,\#1^3-4\,\#1^4+\#1^5\,\&,\,5\right]^3+$$

```
        Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^4),
  y → -√(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5])},
 {x → -√(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5] +
        2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^2 -
        4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^3 +
        Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^4),
  y → √(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5])},
 {x → √(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5] +
       2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^2 -
       4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^3 +
       Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^4),
  y → -√(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5])},
 {x → √(-1 + Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5] +
       2 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^2 -
       4 Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^3 +
       Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5]^4),
  y → √(Root[1 - 2 #1 + #1^2 + 2 #1^3 - 4 #1^4 + #1^5 &, 5])}}
```

Παρατηρούμε ότι η συνάρτηση **Solve** μας επιστρέφει ρίζες χρησιμοποιώντας εκφράσεις της μορφής Root. Στην περίπτωση αυτή μπορούμε να χρησιμοποιήσουμε τη συνάρτηση **NSolve**, η οποία μας επιστρέφει αριθμητικές (προσεγγιστικές) λύσεις:

```
NSolve[sys3, {x, y}]
```

Solve::svars : Equations may not give solutions for all "solve" variables.

```
{{x → -1.48849, y → 0. - 0.906993 i},
 {x → -1.48849, y → 0. + 0.906993 i},
 {x → -0.85559 - 0.684119 i, y → -0.744341 - 0.411259 i},
 {x → -0.85559 - 0.684119 i, y → 0.744341 + 0.411259 i},
 {x → -0.85559 + 0.684119 i, y → -0.744341 + 0.411259 i},
 {x → -0.85559 + 0.684119 i, y → 0.744341 - 0.411259 i},
 {x → -0.838151, y → -1.83339}, {x → -0.838151, y → 1.83339},
 {x → 0. - 0.667931 i, y → -0.831565},
 {x → 0. - 0.667931 i, y → 0.831565},
 {x → 0. + 0.667931 i, y → -0.831565},
 {x → 0. + 0.667931 i, y → 0.831565},
 {x → 0.838151, y → -1.83339}, {x → 0.838151, y → 1.83339},
 {x → 0.85559 - 0.684119 i, y → -0.744341 + 0.411259 i},
 {x → 0.85559 - 0.684119 i, y → 0.744341 - 0.411259 i},
 {x → 0.85559 + 0.684119 i, y → -0.744341 - 0.411259 i},
 {x → 0.85559 + 0.684119 i, y → 0.744341 + 0.411259 i},
 {x → 1.48849, y → 0. - 0.906993 i},
 {x → 1.48849, y → 0. + 0.906993 i}}
```

Κάνουμε επαλήθευση χρησιμοποιούντας τον τελεστή "/." :

```
sys3 /. %
```

```
{{True, True}, {True, True}, {True, True}, {True, True},
 {True, True}, {True, True}, {False, False}, {False, False},
 {True, True}, {True, True}, {True, True}, {True, True},
 {False, False}, {False, False}, {True, True}, {True, True},
 {True, True}, {True, True}, {True, True}, {True, True}}
```

```
{x^2 + 1/y^2, 1/x^4 - 1/y^6} /. %%
```

```
{{1. + 0. i, 2. + 0. i}, {1. + 0. i, 2. + 0. i},
 {1. - 4.44089×10^-16 i, 2. + 3.88578×10^-16 i},
 {1. - 4.44089×10^-16 i, 2. + 3.88578×10^-16 i},
 {1. + 4.44089×10^-16 i, 2. - 3.88578×10^-16 i},
 {1. + 4.44089×10^-16 i, 2. - 3.88578×10^-16 i}, {1., 2.},
 {1., 2.}, {1. + 0. i, 2. + 0. i}, {1. + 0. i, 2. + 0. i},
 {1. + 0. i, 2. + 0. i}, {1. + 0. i, 2. + 0. i}, {1., 2.},
 {1., 2.}, {1. + 4.44089×10^-16 i, 2. - 3.88578×10^-16 i},
 {1. + 4.44089×10^-16 i, 2. - 3.88578×10^-16 i},
 {1. - 4.44089×10^-16 i, 2. + 3.88578×10^-16 i},
 {1. - 4.44089×10^-16 i, 2. + 3.88578×10^-16 i},
 {1. + 0. i, 2. + 0. i}, {1. + 0. i, 2. + 0. i}}
```

Παρατηρούμε, ότι οι λύσεις που επιστρέφει η **NSolve** δεν επαληθεύουν ακριβώς το σύστημα, αλλά τείνουν να το επαληθεύσουν.

Τέλος υπάρχουν συστήματα, στα οποία ούτε η συνάρτηση **NSolve** δεν μπορεί να δώσει προσεγγιστικές λύσεις.

Παράδειγμα 25: Έστω το σύστημα

$$6\,\mathrm{Cos}(x) = y$$
$$x = e^{y}$$

Παρατηρούμε ότι η εξίσωση δεν μπορεί να λυθεί αριθμητικά με τη χρήση της συνάρτησης **NSolve**.

```
NSolve[{6 Cos[x] == y, x == E^y}, {x, y}]
```

```
Solve::tdep : The equations appear to involve the
   variables to be solved for in an essentially non-algebraic way.
```

```
NSolve[{6 Cos[x] == y, x == e^y}, {x, y}]
```

Σε τέτοιες περιπτώσεις, και όχι μόνο σε αυτές, χρησιμοποιούμε τη συνάρτηση **FindRoot**[eqns, {x, x_0}, {y, y_0}], η οποία αναζητεί μια αριθμητική λύση του συστήματος στην περιοχή του σημείου (x, y) = (x_0, y_0). Συγκεκριμένα, αναζητούμε μία αριθμητική λύση στη περιοχή του (x, y) = (***e***, 1):

```
FindRoot[{6 Cos[x] == y, x == E^y}, {x, E}, {y, 1}]
```

```
{x → 1.50285, y → 0.407363}
```

και μία αριθμητική λύση στην περιοχή του σημείου (x, y) = (e^2, 2):

```
FindRoot[{6 Cos[x] == y, x == E^y}, {x, E^2}, {y, 2}]

{x → 7.51124, y → 2.0164}
```

Άσκηση: Προσπαθήστε να βρείτε τις λύσεις με γραφικό τρόπο δηλ. κάνετε τις γραφικές παραστάσεις των συναρτήσεων 6Cos[x] και Log[x] και βρείτε σε ποιά σημεία (x,y) τέμνονται. Για περισσότερες πληροφορίες για το πως σχεδιάζουμε λίστα απο δύο ή παραπάνω γραφικές παραστάσεις μπορείται να διαβάσετε το μάθημα DidiastataGrafhmata.pdf

5.2.6 Απαλοιφή αγνώστων από τις εξισώσεις συστήματος

Το *Mathematica* διαθέτει τη συνάρτηση

Eliminate[{eqns},{vars}]

η οποία μας επιτρέπει να απαλείψουμε τις αναγραφόμενες μέσα στη δεύτερη λίστα μεταβλητές μεταξύ των εξισώσεων του συστήματος.

Παράδειγμα 26: Έστω το σύστημα

$$a\,x + y = 1$$
$$x + (1\text{-}a)\,y = 2$$

Θεωρώντας το a ως παράμετρο και χρησιμοποιώντας τη συνάρτηση **Solve** βρίσκουμε τις ακριβείς λύσεις του συστήματος:

```
Solve[{a x + y == 1, x + (1 - a) y == 2}, {x, y}]
```

$$\left\{\left\{x \to -\frac{-1-a}{1-a+a^2},\ y \to -\frac{-1+2\,a}{1-a+a^2}\right\}\right\}$$

Παρατηρούμε ότι μεταξύ των λύσεων του συστήματος υπάρχει ένας δεσμός(δηλ. μια σχέση που περιγράφεται μαθηματικά με κάποια ή κάποιες εξισώσεις), τον οποίο μπορούμε να βρούμε απαλείφοντας το a μεταξύ των δύο λύσεων. Αυτή ακριβώς, την εργασία κάνει η συνάρτηση **Eliminate**.

Εύρεση του δεσμού με τη χρήση της συνάρτησης **Eliminate**:

```
eqn2 = Eliminate[{a x + y == 1, x + (1 - a) y == 2}, a]

(1 - y) y == x^2 + x (-2 + y)
```

Αν θέλουμε τώρα να επιλύσουμε τον δεσμό ως προς μία εκ των δύο μεταβλητών, μπορούμε ασφαλώς να χρησιμοποιήσουμε τη συνάρτηση **Solve**.

Επίλυση του δεσμού ως προς τη μεταβλητή x:

```
Solve[eqn2, x]
```

$$\left\{\left\{x \to \frac{1}{2}\left(2 - y - \sqrt{4 - 3\,y^2}\right)\right\}, \left\{x \to \frac{1}{2}\left(2 - y + \sqrt{4 - 3\,y^2}\right)\right\}\right\}$$

Επίλυση του δεσμού ως προς τη μεταβλητή y:

```
Solve[eqn2, y]
```

$$\left\{\left\{y \to \frac{1}{2}\left(1 - x - \sqrt{1 + 6\,x - 3\,x^2}\right)\right\}, \left\{y \to \frac{1}{2}\left(1 - x + \sqrt{1 + 6\,x - 3\,x^2}\right)\right\}\right\}$$

Παρατήρηση: Η συνάρτηση Solve μας παρέχει τη δυνατότητα να καταλήξουμε στο ίδιο αποτελέσματα. Συγκεκριμένα:

αν θέλουμε να λύσουμε το σύτημα ως προς τη μεταβλητή x απαλείφοντας την παράμετρο a, χρησιμοποιούμε την εντολή:

```
Solve[{a x + y == 1, x + (1 - a) y == 2}, x, a]
```

$$\left\{\left\{x \to \frac{1}{2}\left(2 - y - \sqrt{4 - 3\,y^2}\right)\right\}, \left\{x \to \frac{1}{2}\left(2 - y + \sqrt{4 - 3\,y^2}\right)\right\}\right\}$$

ενώ αν θέλουμε να λύσουμε το σύτημα ως προς τη μεταβλητή y απαλείφοντας την παράμετρο a, χρησιμοποιούμε την εντολή:

```
Solve[{a x + y == 1, x + (1 - a) y == 2}, y, a]
```

$$\left\{\left\{y \to \frac{1}{2}\left(1 - x - \sqrt{1 + 6\,x - 3\,x^2}\right)\right\}, \left\{y \to \frac{1}{2}\left(1 - x + \sqrt{1 + 6\,x - 3\,x^2}\right)\right\}\right\}$$